
Abstract 
•  RPG Open Access (RPGOA) has been called "the future of RPG on 

IBM i." Yet we have been told there are myths surrounding RPGOA. 
A standards committee has been established to guide the community 
to a consistent approach to RPGOA development and 
communication, but it has yet to gain traction. Looking back to the 
introduction of RPGOA, the confusion and lack of information 
persists. 

•  This session will explain RPG Open Access, starting with a basic 
explanation and overview. It will describe several actual customer 
implementations, and theorize on future possible uses for RPGOA. 
This session will also introduce you to resources on how to build your 
own RPGOA handlers to extend your existing applications and build 
modern IBM i applications from scratch.  

What? 
•  IBM® Rational® Open Access: RPG Edition 
http://www-01.ibm.com/software/rational/products/openaccess/ 

Rational Open Access RPG Edition 
•  Access new devices and resources 

with extensions for RPG applications 
–  IBM® Rational® Open Access RPG Edition provides the runtime 

enablement for the I/O handlers that are developed to extend ILE RPG 
I/O capabilities to new resources and devices. 

–  It allows programmers to use the simple, well-understood RPG I/O 
model to access new devices and resources that are not directly 
supported by RPG. Rational Open Access RPG Edition is a full-
featured version of the limited support provided by native RPG 
SPECIAL files.  

Rational Open Access RPG Edition 
software: 
•  Accesses multiple devices and resources 

–  such as mobile devices, web services, cloud resources, external databases, XML files and 
spreadsheets. 

•  Supports existing handlers provided by third parties 
–  including software vendors, business partners, service organizations and open-source 

software. 

•  Takes advantage of existing RPG programming skills 
–  by enabling RPG developers to access the resource or device in their applications using 

native RPG file specifications and I/O operations. 

•  Supports RPG IV language enhancements 
–  including the addition of ILE, which enables you to write modular, reusable code following 

programming best practices. 

•  Is a technology that is now included at no additional charge 
–  with the IBM i RPG compiler and runtime (Rational Open Access RPG Edition was formerly 

a priced, keyed offering) 

My First Experience 
•  Customer using a refacing/repurposing tool 

–  5250 for OS and existing programs 
–  OA version for handler 

•  Decided on RPG Open Access solution 
•  Found a recently written program 

–  Installed the handler 
•  Started the solution 
•  Signed on 
•  Navigated to the program 
•  Nothing looks different! 

5250 refaced 

RPG OA 

What happened? 
•  Subfile delivered to the client 

–  5250: One page at a time 
–  RPGOA: The entire list 

•  Client’s built in grid sorting functionality 
–  5250: One page at a time 
–  RPGOA: The entire list 

•  Date fields turned into combos 

5250 refaced 

RPG OA 

5250 refaced 

RPG OA 

5250 refaced 

RPG OA 

What was the benefit? 
•  RPG code to sort the list no longer needed 

– Reduced maintenance 

•  Response time – different paradigm 

What was the cost? 
•  RPG code to sort the list no longer needed 

•  Existing RPG programs did not work with OA handler 
– One-page-replace subfile coding technique 

Anti-climax! 
•  ONE line of code.. 
FDisplay   UF   E             WORKSTN 

F                             HANDLER(‘handler-program’) 

5250 Display 

Webfacing APIs 

RPG Open Access Handler 

What else? 
•  Drop down combo boxes 

–  List built inside RPG program 
•  Maps, Images, Links 

– Collect data from inside RPG program 

•  More data 
– Not limited to the size of 24x80 or 27x132 data stream 

What else? 
•  Separation of application and user interface 

– Don’t need data collection at glass 
– Extend the application with existing RPG & SQL skills 

•  How?  
– Fields added to the DDS 

•  Can be H(idden) fields 
•  Handler can manage more data/fields 

– DDS directive connects fields to GUI properties 

Who does it? 
•  Replace 5250 data stream 

– Profound 
–  looksoftware 
– ASNA 
– Rocket 

That’s it? 
•  NO! 

•  The power of RPG Open Access is purported to be 
in replacing the 5250 data stream 

•  Vendors have focused on user interface 
•  Press has focused on vendor focus 
•  Vendors are suggesting there are myths 

Rational Open Access RPG Edition 
•  Access new devices and resources 

with extensions for RPG applications 
–  IBM® Rational® Open Access RPG Edition provides the runtime 

enablement for the I/O handlers that are developed to extend ILE RPG 
I/O capabilities to new resources and devices. 

–  It allows programmers to use the simple, well-understood RPG I/O 
model to access new devices and resources that are not directly 
supported by RPG. Rational Open Access RPG Edition is a full-
featured version of the limited support provided by native RPG 
SPECIAL files.  

RPG Open Access Handler 

So… 
•  RPG Open Access cannot replace 5250 completely 

– Operating system screens 
– Older green screen applications 

•  No source 
•  No maintenance allowed 

•  RPG Open Access for 5250 must be hybrid 
•  Building your own is complicated 

– Can be done, but… 
–  Individual or generic? 

Who does it? – part 2! 
•  Replace 5250 data stream - devices 

– Profound 
–  looksoftware 
– ASNA 
– Rocket 

•  Something else? - resources 
– RJS 
– Rocket 

RJS Special Report 

RJS Special Report 
•  With RPG OA, RPG programmers are able to use simple and 

familiar RPG I/O operations to access devices and resources 
that RPG was not originally designed to support using its I/O 
model.  
– These resources include: 

•  Browsers 
•  Mobile devices 
•  Web services 
•  XML files 
•  External databases 
•  Spreadsheets 
•  Devices and resources yet unknown 

RJS Special Report 
•  The design of an Open Access application starts 

with these elements: 
– An RPG program that uses standard I/O code 
– A resource or device which the RPG program will utilize in some way 

•  The component that binds these elements together 
is the handler. 

RJS OA Handler – RPG2SQL 

What I did 
•  Gave up one evening 
•  Researched online 
•  Wrote two RPG programs 

– Database file to printer 
– Database file to database file 

•  Wrote two handlers 
– Create HTML data 
– Create XML data 

Resources 
•  Home page 

–  http://www-01.ibm.com/software/rational/products/openaccess 

•  Prerequisites 
–  RPG Open Access 

•  http://tinyurl.com/rpg‐oar‐ptfs  
–  DB2 For i latest group PTFs 

•  https://www.ibm.com/developerworks/ibmi/techupdates/db2  

•  Developer Works – RPG Café 
–  http://tinyurl.com/rpg-open-access 

Documentation 
http://pic.dhe.ibm.com/infocenter/iseries/v7r1m0/topic/books/rzasm.pdf 

IBM i

Rational Open Access: RPG Edition

Resources 
•  IBM Systems Magazine – RPG 

–  http://www.ibmsystemsmag.com/ibmi/developer/rpg 
–  Jon Paris and Susan Gantner 

–  July 2010: Getting a Handle on RPG’s Open Access 
•  http://partner400.com/examples/openaccessexample1.htm 

–  November 2012: RPG Open Access Revisited 
–  December 2012: Implementing Open Access With RPG 
–  January 2013: Handling Input Handlers With RPG Open Access 

Resources 
•  iPro Developer 

–  Rational Open Access: RPG Edition Workshop 
–  Craig Pelkie 
–  January 2013: Part 1 

•  http://iprodeveloper.com/application-development/rational-open-access-rpg-
edition-workshop-part-1 

–  March 2013: Part 2 
•  http://iprodeveloper.com/application-development/rational-open-access-rpg-

edition-workshop-part-2 
–  May 2013: Part 3 

•  http://iprodeveloper.com/rpg-programming/rational-open-access-rpg-
edition-workshop-part-3 

Movies DB2 file 
fmovie     IF   E             DISK 

flist001   O    E             PRINTER 

/free 

 read movier; 

 dow not %eof; 

   write outfmt; 

   read movier; 

 enddo; 

 *inlr = *on; 

 /end-free  

39 

Movies Application 

List program 1 
fmovie     IF   E             DISK 

flist001   O    E             PRINTER 

/free 

 read movier; 

 dow not %eof; 

   write outfmt; 

   read movier; 

 enddo; 

 *inlr = *on; 

 /end-free  

41 

List printer file 1 
A                                      REF(MOVIE) 

A          R OUTFMT                    SPACEB(1) 

A            MOVNAME   R              1 

A            MOVIMG    R             +2  

List 1 output 
 Display Spooled File                               

File  . . . . . :   LIST001                          Page/Line   1/2            

Control . . . . .                                    Columns     1 - 78         

Find  . . . . . .                                                               

*...+....1....+....2....+....3....+....4....+....5....+....6....+....7....+...  

Crocodile Dundee                                    CrocodileDundee.jpg         

William Shakespeare                                 Shakespeare.jpg             

Ken Follett                                         KenFollett.jpg              

Leonardo da Vinci                                   LeonardoDaVinci.jpg         

Harry Potter                                        HarryPotter.jpg             

Toy Story 1                                         ToyStory1.jpg               

Toy Story 2                                         ToyStory2.jpg               

Toy Story 3                                         ToyStory3.jpg               

Mike wants you                                      Mike.jpg                    

Cloudy with a Chance of Meatballs                   Cloudy.jpg                  

Shrek                                               Shrek.jpg                   

Shrek 2                                             Shrek2.jpg                  

Shrek The Third                                     Shrek3.jpg                  

Shrek Forever After                                 ShrekForeverAfter.jpg       

Robin Hood                                          RobinHood.jpg               

The Princess Bride                                  ThePrincessBride.jpg        

                                                                       More...  

F3=Exit   F12=Cancel   F19=Left   F20=Right   F24=More keys  

List program 2 
fmovie     IF   E             DISK 

flist002   O    E             DISK 

/free 

 read movier; 

 dow not %eof; 

   write outfmt; 

   read movier; 

 enddo; 

 *inlr = *on; 

 /end-free  

List DB2 output file 2 
A                                      REF(MOVIE) 

A          R OUTFMT 

A            MOVNAME   R 

A            MOVIMG    R  

45 

List 1 output 
 Display Report                                  

                                              Report width . . . . . :     102   

 Position to line  . . . . .              Shift to column  . . . . . .           

 Line   ....+....1....+....2....+....3....+....4....+....5....+....6....+....7.. 

        Movie Name                                          Cover img            

 000001 Crocodile Dundee                                    CrocodileDundee.jpg  

 000002 William Shakespeare                                 Shakespeare.jpg      

 000003 Ken Follett                                         KenFollett.jpg       

 000004 Leonardo da Vinci                                   LeonardoDaVinci.jpg  

 000005 Harry Potter                                        HarryPotter.jpg      

 000006 Toy Story 1                                         ToyStory1.jpg        

 000007 Toy Story 2                                         ToyStory2.jpg        

 000008 Toy Story 3                                         ToyStory3.jpg        

 000009 Mike wants you                                      Mike.jpg             

 000010 Cloudy with a Chance of Meatballs                   Cloudy.jpg           

 000011 Shrek                                               Shrek.jpg            

 000012 Shrek 2                                             Shrek2.jpg           

 000013 Shrek The Third                                     Shrek3.jpg           

 000014 Shrek Forever After                                 ShrekForeverAfter.jp 

 000015 Robin Hood                                          RobinHood.jpg        

 000016 The Princess Bride                                  ThePrincessBride.jpg 

                                                                      More...    

 F3=Exit      F12=Cancel      F19=Left      F20=Right      F21=Split  

DB2 output file for HTML 
CREATE TABLE OALIB.HTMLOUT 

 ( HTML VARCHAR(1024) CCSID 37 NOT NULL DEFAULT '' ) 

 RCDFMT HTMLFMT; 

 

LABEL ON TABLE OALIB.HTMLOUT 

 IS 'HTML OUTPUT FILE‘;  

DB2 output file for XML 
CREATE TABLE OALIB.XMLOUT 

 ( XML VARCHAR(1024) CCSID 37 NOT NULL DEFAULT '' ) 

 RCDFMT XMLFMT; 

 

LABEL ON TABLE OALIB.XMLOUT 

 IS ‘XML OUTPUT FILE‘;  

Handler for HTML 
QOA.useNamesValues = '1'; 

pQOAdata = QOA.namesValues; 

HTML +=tagRowStart; 

For fieldNum = 1 to QOAdata.num; 

 pFieldValue = QOAdata.field(fieldNum).value; 

 fieldLength = QOAdata.field(fieldNum).valueLenBytes; 

 HTML += tagColStart 

      + %trimr(%subst(fieldValue: 1: fieldLength)) 

      + tagColEnd; 

EndFor; 

HTML += tagRowEnd; 

write htmlfmt; 

Handler for XML 
QOA.useNamesValues = '1'; 

pQOAdata = QOA.namesValues; 

XML = tagRowStart; 

For fieldNum = 1 to QOAdata.num; 

  pFieldValue = QOAdata.field(fieldNum).value; 

  fieldName = QOAdata.field(fieldNum).externalName; 

  fieldLength = QOAdata.field(fieldNum).valueLenBytes; 

  XML += tagElmStart + %trimr(fieldName) + tagElmBack 

      + %trimr(%subst(fieldValue: 1: fieldLength)) 

      + tagElmEnd + %trimr(fieldName) + tagElmBack; 

EndFor; 

XML += tagRowEnd; 

write XMLfmt;  

List program 1 
Fmovie     IF   E             DISK 

Flist001   O    E             PRINTER 

F                             handler('HDLR_HTML') 

/free 

 read movier; 

 dow not %eof; 

   write outfmt; 

   read movier; 

 enddo; 

 *inlr = *on; 

 /end-free  

List program 2 
Fmovie     IF   E             DISK 

Flist001   O    E             DISK 

F                             handler('HDLR_XML') 

/free 

 read movier; 

 dow not %eof; 

   write outfmt; 

   read movier; 

 enddo; 

 *inlr = *on; 

 /end-free  

XMLOUT 
<movies>                                                                 

  <movie> 

    <MOVNAME>Crocodile Dundee</MOVNAME> 

    <MOVIMG>CrocodileDundee.jpg</MOVIMG> 

  </movie> 

  <movie> 

    <MOVNAME>William Shakespeare</MOVNAME> 

    <MOVIMG>Shakespeare.jpg</MOVIMG> 

 </movie> 

  <movie> 

    <MOVNAME>Ken Follett</MOVNAME> 

    <MOVIMG>KenFollett.jpg</MOVIMG> 

  </movie> 

... 

</movies>                                                                 

HTMLOUT 
<table>                                                                

<thead><tr><th>MOVNAME</th><th>MOVIMG</th></tr></thead>                

<tbody><tr><td>Crocodile Dundee</td><td>CrocodileDundee.jpg</td></tr>  

<tr><td>William Shakespeare</td><td>Shakespeare.jpg</td></tr>          

<tr><td>Ken Follett</td><td>KenFollett.jpg</td></tr>                   

<tr><td>Leonardo da Vinci</td><td>LeonardoDaVinci.jpg</td></tr>        

<tr><td>Harry Potter</td><td>HarryPotter.jpg</td></tr>                 

<tr><td>Toy Story 1</td><td>ToyStory1.jpg</td></tr>                    

<tr><td>Toy Story 2</td><td>ToyStory2.jpg</td></tr>                    

<tr><td>Toy Story 3</td><td>ToyStory3.jpg</td></tr>                    

<tr><td>Mike wants you</td><td>Mike.jpg</td></tr> 

... 

</tbody></table>  

HTMLOUT 

XML Handler description 
•  Parameters 
•  Operation 
•  Field names/values 

Entry parameters 
 /copy QOAR/QRPGLESRC,QRNOPENACC 

 

D QOA             ds                  likeds(QrnOpenAccess_T) 

 

C     *ENTRY        PLIST 

C                   PARM                    QOA  

Operation 
 /copy QOAR/QRPGLESRC,QRNOPENACC 

 

D QOA             ds                  likeds(QrnOpenAccess_T) 

D opcode          s                   like(QOA.rpgOperation) 

 

 opcode = QOA.rpgOperation; 

 if opcode = QrnOperation_OPEN; 

   open XMLOUT; 

   XML = tagTableStart; 

   write XMLfmt; 

   return; 

EndIf; 

Operations 
•  QrnOperation_CHAIN 
•  QrnOperation_CLOSE 
•  QrnOperation_DELETE 
•  QrnOperation_EXFMT 
•  QrnOperation_FEOD 
•  QrnOperation_OPEN 
•  QrnOperation_READ 
•  QrnOperation_READC 
•  QrnOperation_READE 
•  QrnOperation_READP 

•  QrnOperation_READPE 
•  QrnOperation_SETGT 
•  QrnOperation_SETLL 
•  QrnOperation_UNLOCK 
•  QrnOperation_UPDATE 
•  QrnOperation_WRITE 
•  QrnOperation_DELETE_CURRENT 
•  QrnOperation_READE_CURRENT 
•  QrnOperation_READPE_CURRENT 
•  QrnOperation_POSITION_START 
•  QrnOperation_POSITION_END 

Documentation 

60 

http://pic.dhe.ibm.com/infocenter/iseries/v7r1m0/topic/books/rzasm.pdf 

Rational Open Access: RPG Edition

Field Names/Values 
 /copy QOAR/QRPGLESRC,QRNOPENACC 

D QOA             ds                  likeds(QrnOpenAccess_T) 

D QOAdata         ds                  likeds(QrnNamesValues_T) 

D                                     based(pQOAdata) 

 

QOA.useNamesValues = '1'; 

pQOAdata = QOA.namesValues; 

XML = tagRowStart; 

For fieldNum = 1 to QOAdata.num; 

  pFieldValue = QOAdata.field(fieldNum).value; 

  fieldName   = QOAdata.field(fieldNum).externalName; 

  fieldLength = QOAdata.field(fieldNum).valueLenBytes; 

  XML += tagElmStart + %trimr(fieldName) + tagElmBack 

      + %trimr(%subst(fieldValue: 1: fieldLength)) 

      + tagElmEnd + %trimr(fieldName) + tagElmBack; 

EndFor; 

What next? 
•  Write some handler code! 

– Output 
–  Input 

•  Make some decisions 
– Generic handler 
– Specific handler 

•  Investigate 
– Third-party handlers 

XML Handler 
     H option(*srcstmt : *nodebugio) 

     H dftactgrp(*no) 

 

     fxmlout    O    E             DISK    usropn 

 

      /copy QOAR/QRPGLESRC,QRNOPENACC 

 

     D QOA             ds                  likeds(QrnOpenAccess_T) 

     D QOAdata         ds                  likeds(QrnNamesValues_T) 

     D                                     based(pQOAdata) 

     D opcode          s                   like(QOA.rpgOperation) 

     D fieldValue      s          32470a   Based(pFieldValue) 

     D fieldName       s             10 

     D fieldNum        s             10  0 

     D fieldLength     s             10  0 

XML Handler 
     d tagTableStart   c                   '<movies>' 

     d tagTableEnd     c                   '</movies>' 

     d tagRowStart     c                   '<movie>' 

     d tagRowEnd       c                   '</movie>' 

     d tagElmStart     c                   '<' 

     d tagElmEnd       c                   '</' 

     d tagElmBack      c                   '>' 

 

     C     *ENTRY        PLIST 

     C                   PARM                    QOA 

 

      /free 

 

       XML  = ''; 

       QOA.useNamesValues = '1’; 

XML Handler 
       opcode = QOA.rpgOperation; 

       if opcode = QrnOperation_WRITE; 

         pQOAdata = QOA.namesValues; 

         XML  = tagRowStart; 

         For fieldNum = 1 to QOAdata.num; 

           pFieldValue = QOAdata.field(fieldNum).value; 

           fieldName   = QOAdata.field(fieldNum).externalName; 

           fieldLength = QOAdata.field(fieldNum).valueLenBytes; 

           XML  += tagElmStart + %trimr(fieldName) + tagElmBack 

                + %trimr(%subst(fieldValue: 1: fieldLength)) 

                + tagElmEnd + %trimr(fieldName) + tagElmBack; 

         EndFor; 

         XML  += tagRowEnd; 

         write XMLfmt; 

         return; 

       EndIf; 

XML Handler 
       if opcode = QrnOperation_OPEN; 

         open XMLOUT; 

         XML  = tagTableStart; 

         write XMLfmt; 

         return; 

       EndIf; 

       if opcode = QrnOperation_CLOSE; 

         XML  = tagTableEnd; 

         write XMLfmt; 

         close XMLOUT; 

         return; 

       EndIf; 

 

       return; 

      /end-free  

